Shader "Unlit/Instanced2Shader"
{
    Properties
    {
        _Color("Color",Color) = (1,1,1,1)
        _MainTex("MainTex",2D) = "gray"{}
    }

    SubShader
    {
        Tags
        {
            "RenderPipeline"="UniversalPipeline"
            "RenderType"="Opaque"
            "Queue"="Geometry+0"
        }

        Pass
        {
            Name "Forward"
			Tags
            { 
			    "LightMode"="UniversalForward"
			}
            
            Blend One Zero,One Zero
            Cull Back
            ZTest LEqual
            ZWrite On

            HLSLPROGRAM
            #pragma vertex vert
            #pragma fragment frag
            #pragma target 4.5

            #include "Packages/com.unity.render-pipelines.core/ShaderLibrary/Color.hlsl"
            #include "Packages/com.unity.render-pipelines.universal/ShaderLibrary/Core.hlsl"
            #include "Packages/com.unity.render-pipelines.universal/ShaderLibrary/Lighting.hlsl"

            CBUFFER_START(UnityPerMaterial)
            half4 _Color;
            float4 _MainTex_ST;
            CBUFFER_END

            TEXTURE2D(_MainTex);

            //为了方便操作 定义预定义
            #define smp SamplerState_Point_Repeat
            // SAMPLER(sampler_MainTex); 默认采样器
            SAMPLER(smp);

            StructuredBuffer<float4x4> transBuf; // 定义接收的buff
            
            // 顶点输入
            struct Attributes
            {
                // float4 positionOS : POSITION; // 位置直接进行矩阵计算时，需要是 float4 的格式，如果是 float3 和 float4x4 运算时会丢失最下面的 translate 位移信息，导致没有位移，位置为0,0,0
                float3 positionOS : POSITION; // 位置是 float3 时，进行矩阵计算时，要转为 float4 格式，原因同上
                float2 uv : TEXCOORD0;
                uint instanceID : SV_InstanceId; // instanceID 实例在缓冲中的索引
            };

            //顶点输出
            struct Varyings
            {
                float4 positionCS : SV_POSITION;
                float2 uv : TEXCOORD0;
            };

            Varyings vert(Attributes v)
            {
                float4x4 trans = transBuf[v.instanceID]; // 取出缓冲中的矩阵数据
                
                Varyings o;

                // 直接取矩阵内部数据
                // const float x = trans[0][3];
                // const float y = trans[1][3];
                // const float z = trans[2][3];
                // float3 pos = float3(x, y, z);
                // o.positionCS = TransformWorldToHClip(v.positionOS + pos);

                // o.positionCS = TransformWorldToHClip(mul(trans, v.positionOS)); // positionOS 为 float4 时，直接运算即可
                o.positionCS = TransformWorldToHClip(mul(trans, float4(v.positionOS.xyz, 1))); // positionOS 为 float3 时，需要变换为 float4 ，注意最后一个 w 要设置为 1
                
                o.uv = TRANSFORM_TEX(v.uv, _MainTex);
                
                return o;
            }

            half4 frag(Varyings i) : SV_TARGET
            {
                half4 mainTex = SAMPLE_TEXTURE2D(_MainTex, smp, i.uv);
                half4 c = _Color * mainTex;
                return c;
            }
            ENDHLSL
        }
    }
}
